NAG Library Routine Document

G01EDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01EDF returns the probability for the lower or upper tail of the F or variance-ratio distribution with real degrees of freedom, via the routine name.

2 Specification

```
FUNCTION G01EDF (TAIL, F, DF1, DF2, IFAIL)
REAL (KIND=nag_wp) G01EDF
INTEGER            IFAIL
REAL (KIND=nag_wp) F, DF1, DF2
CHARACTER(1)       TAIL
```

3 Description

The lower tail probability for the F, or variance-ratio distribution, with ν_1 and ν_2 degrees of freedom, $P(F \leq f : \nu_1, \nu_2)$, is defined by:

$$P(F \leq f : \nu_1, \nu_2) = \frac{\nu_1^{\nu_1/2}\nu_2^{\nu_2/2}\Gamma((\nu_1+\nu_2)/2)}{\Gamma(\nu_1/2)\Gamma(\nu_2/2)}\int_0^f F^{(\nu_1-2)/2}(\nu_1 F+\nu_2)^{-(\nu_1+\nu_2)/2}dF,$$

for $\nu_1,\ \nu_2 > 0,\ f \geq 0$.

The probability is computed by means of a transformation to a beta distribution, $P_\beta(B \leq \beta : a, b)$:

$$P(F \leq f : \nu_1, \nu_2) = P_\beta\left(B \leq \frac{\nu_1 f}{\nu_1 f + \nu_2} : \nu_1/2, \nu_2/2\right)$$

and using a call to G01EEF.

For very large values of both ν_1 and ν_2, greater than 10^5, a normal approximation is used. If only one of ν_1 or ν_2 is greater than 10^5 then a χ^2 approximation is used, see Abramowitz and Stegun (1972).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: TAIL – CHARACTER(1) *Input*

On entry: indicates whether an upper or lower tail probability is required.

TAIL = 'L'
The lower tail probability is returned, i.e., $P(F \leq f : \nu_1, \nu_2)$.

TAIL = 'U'
The upper tail probability is returned, i.e., $P(F \geq f : \nu_1, \nu_2)$.

Constraint: TAIL = 'L' or 'U'.

2: F – REAL (KIND=nag_wp) *Input*

On entry: f, the value of the F variate.

Constraint: F $\geq$ 0.0.

3: DF1 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the numerator variance, ν_1.

Constraint: DF1 $>$ 0.0.

4: DF2 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the denominator variance, ν_2.

Constraint: DF2 $>$ 0.0.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01EDF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If IFAIL $= 1$, 2 or 3 on exit, then G01EDF returns 0.0.

IFAIL $= 1$

On entry, TAIL $\neq$ 'L' or 'U'.

IFAIL $= 2$

On entry, F $<$ 0.0.

IFAIL $= 3$

On entry, DF1 $\leq$ 0.0,
or DF2 $\leq$ 0.0.

IFAIL $= 4$

F is too far out into the tails for the probability to be evaluated exactly. The result tends to approach 1.0 if f is large, or 0.0 if f is small. The result returned is a good approximation to the required solution.

7 Accuracy

The result should be accurate to five significant digits.

8 Further Comments

For higher accuracy G01EEF can be used along with the transformations given in Section 3.

9 Example

This example reads values from, and degrees of freedom for, a number of F-distributions and computes the associated lower tail probabilities.

9.1 Program Text

```
    Program g01edfe

!     G01EDF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g01edf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: df1, df2, f, prob
      Integer                          :: ifail
      Character (1)                    :: tail
!     .. Executable Statements ..
      Write (nout,*) 'G01EDF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) ' TAIL      F      DF1     DF2     Probability'
      Write (nout,*)

d_lp: Do
        Read (nin,*,Iostat=ifail) tail, f, df1, df2
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Calculate probability
        ifail = -1
        prob = g01edf(tail,f,df1,df2,ifail)
        If (ifail/=0) Then
          If (ifail/=4) Then
            Exit d_lp
          End If
        End If

!       Display results
        Write (nout,99999) tail, f, df1, df2, prob
      End Do d_lp

99999 Format (3X,A1,4X,F6.3,2F8.2,7X,F7.4)
    End Program g01edfe
```

9.2 Program Data

```
G01EDF Example Program Data
'L'  5.5   1.5   25.5 : TAIL F DF1 DF2
'L' 39.9   1.0    1.0
'L' 2.5   20.25   1.0
```

9.3 Program Results

```
 G01EDF Example Program Results

  TAIL      F       DF1     DF2     Probability

   L     5.500     1.50   25.50        0.9837
   L    39.900     1.00    1.00        0.9000
   L     2.500    20.25    1.00        0.5342
```